

Encoding Higher Inductive Types Without Boilerplate

A Study in Agda Metaprogramming

Paventhan Vivekanandan
School of Informatics, Computing and Engineering
Indiana University
Bloomington, Indiana, USA
pvivekan@indiana.edu

David Thrane Christiansen
Galois, Inc.
Portland, Oregon, USA
dtt@galois.com

Abstract

Higher inductive types are inductive types that include non-trivial higher-dimensional structure, represented as identifications at the type that are not reflexivity. While work proceeds on type theories with a computational interpretation of univalence and higher inductive types, it is convenient to encode these structures in more traditional type theories. However, these encodings involve a great deal of error-prone additional syntax. We present a library that uses Agda's metaprogramming facilities to automate this process, allowing higher inductive types to be specified with minimal additional syntax.

Keywords Higher inductive type, Elaboration, Elimination rules, Computation rules

1 Introduction

A metaprogram is a program used to create other programs. In dependently-typed programming languages like Agda and Coq, a reflection library provides metaprogramming interfaces to support code generation for proofs and programs. Elaborator reflection [8], a new paradigm for metaprogramming, provides the language users with a powerful set of interfaces for efficient code generation. An *elaborator* is a metaprogram that converts a high-level language syntax in its abstract representation to a core language which is then type checked independently. By exposing a primitive monad to elaboration framework in Idris, Christiansen [8] showed how to write termination proofs for general recursive functions by automating Bove-Capretta [5] transformation, and how to generate data types matching schemas of external data sources based on Idris's type providers [6].

Agda is a proof-assistant based on Martin-Löf's intensional type theory. Unlike Idris, which has a dedicated elaborator to convert the high-level syntax to its core language, Agda's elaboration mechanism is a part of its type checker. Inspired by Idris's elaborator reflection, Agda development team extended the reflection library of Agda by exposing its elaboration monad to its high-level metaprogramming instructions. The elaboration monad provides an interface

to the Agda type checker through a set of primitive operations which can be used to retrieve static type information regarding various code segments. The primitive operations can also be used to build code fragments using constructs of an abstract syntax tree, and to convert an abstract syntax tree to its concrete syntax.

In this paper, we discuss metaprogramming using elaborator reflection in Agda. By using the elaboration monad, we performed code generation for the elimination and computation rules of inductive types and higher inductive types. We also automated the code generation for the boiler-plate code segment used to define the higher inductive type. We discuss the construction process in detail and introduce new metaprogramming interfaces which build the code segments using elaborator reflection. The interfaces extensively use Agda's reflection primitives and build the abstract syntax tree of the code segments with static type information obtained using the reflection primitives. The generated code is then brought into scope by another top-level reflection primitive.

Elaborator reflection is a powerful mechanism which allows us to create more interesting metaprograms. By performing the above automation, we abstracted the difficulties involved in implementing and using inductive types and higher inductive types by automating the construction of their corresponding elimination and computation rules. We also achieved an extensive reduction in code size, and we demonstrated how to extend a language without modifying the compiler or depending on the language implementers. More specifically, we discuss the following contributions in this paper.

- We automate the construction of the recursion and the induction principles for inductive types with constructors taking zero arguments, one or more arguments, and the type being defined itself as an argument.
- We discuss the automation of code generation of boiler-plate code for a higher inductive type defined inside a module. The boiler-plate code depends on a base type defined as private inside the module. The constructors of the private base type are not accessible outside the module.

Insert a n
of Licata'
that we're
ing

Rewrite c
tions list
on actual
things - n
ple have
eliminato
tion before

- We discuss the automation of code generation of the elimination and the computation rules for a higher inductive type with point and path constructors. We demonstrate the code generation of the reduction rules, specified as postulates, for the path constructors of the higher inductive type.
- We discuss the automation of code generation for patch theory [4] implementation enriched with patches of encryption. We also examine the usage of the automation tool in a cryptography application [18].

The automation tool abstracts the implementation difficulties of a higher inductive type and its recursion and induction principles. We automated the code generation of approximately 1500 lines of code of the patch theory implementation with just 70 lines of automation code. The automation code extensively uses the static type information queried from the type checker using the reflection primitives.

2 Background

2.1 Agda Reflection

Agda's reflection library enables compile-time metaprogramming. This reflection library directly exposes parts of the implementation of Agda's type checker and elaborator for use by metaprograms, in a manner that is similar to Idris's elaborator reflection [7, 8] and Lean's tactic metaprogramming [10]. The type checker's implementation is exposed as effects in a monad called TC.

Agda exposes a representation of its syntax to metaprograms, including datatypes for expressions (called *Term*) and definitions (called *Definition*). The primitives exposed in TC include declaring new metavariables, unifying two *Terms*, declaring new definitions, adding new postulates, computing the normal form or weak head normal form of a *Term*, inspecting the current context, and constructing fresh names. This section describes the primitives that are used in our code generation library; more information on the reflection library can be found in the Agda documentation [1].

TC computations can be invoked in three ways: by macros, which work in expression positions, using the `unquoteDecl` operator in a declaration position, which can bring new names into scope, and using the `unquoteDef` operator in a declaration position, which can automate constructions using names that are already in scope. This preserves the principle in Agda's design that the system never invents a name.

An Agda *macro* is a function of type $t_1 \rightarrow t_2 \rightarrow \dots \rightarrow \text{Term} \rightarrow \text{TC } \top$ that is defined inside a `macro` block. Macros are special: their last argument is automatically supplied by the type checker, and consists of a *Term* that represents the metavariable to be solved by the macro. If the remaining arguments are quoted names or *Terms*, then the type checker will automatically quote the arguments at the macro's use

```
macro
  mc1 : Term → Term → TC ⊤
  mc1 exp hole =
    do exp' ← quoteTC exp
    unify hole exp'
```

```
sampleTerm : Term
sampleTerm = mc1 (λ (n : Nat) → n)
```

Figure 1. A macro that quotes its argument

```
macro
  mc2 : Term → Term → TC ⊤
  mc2 exp hole =
    do exp' ← unquoteTC exp
    unify hole exp'
```

```
sampleSyntax : Nat → Nat
sampleSyntax =
  mc2 (lam visible (abs "n" (var 0 [])))
```

Figure 2. A macro that unquotes its argument

site. At some point, the macro is expected to unify the provided metavariable with some other term, thus solving it.

Figure 1 demonstrates a macro that quotes its argument. The first step is to quote the quoted expression argument again, using `quoteTC`, yielding a quotation of a quotation. The result of this double-quotation is passed, using Agda's new support for Haskell-style `do`-notation, into a function that unifies it with the hole. Because unification removes one layer of quotation, `unify` inserts the original quoted term into the hole. The value of `sampleTerm` is

```
lam visible (abs "n" (var 0 []))
```

The constructor `lam` represents a lambda, and its body is formed by the abstraction constructor `abs` that represents a scope in which a new name "n" is bound. The body of the abstraction is a reference back to the abstracted name using de Bruijn index 0.

The `unquoteTC` primitive removes one level of quotation. Figure 2 demonstrates the use of `unquoteTC`. The macro `mc2` expects a quotation of a quotation, and substitutes its unquotation for the current metavariable.

The `unquoteDecl` and `unquoteDef` primitives, which run TC computations in a declaration context, will typically introduce new declarations by side effect. A function of a given type is declared using `declareDef`, and it can be given a definition using `defineFun`. Similarly, a postulate of a given type is defined using `declarePostulate`. Figure 3 shows an Agda implementation of addition on natural numbers, while figure 4 demonstrates an equivalent metaprogram that adds the same definition to the context.

In Figure 4, `declareDef` declares the type of `plus`. The constructor `pi` represents dependent function types, but a pattern synonym is used to make it shorter. Similarly, `def` constructs references to defined names, and the pattern

```
plus : Nat → Nat → Nat
plus zero b = b
plus (suc n) b = suc (plus n b)
```

Figure 3. Addition on natural numbers

```
pattern vArg x = arg (arg-info visible relevant) x
pattern _`⇒_ a b = pi (vArg a) (abs "_" b)
pattern `Nat = def (quote Nat) []

unquoteDecl plus =
  do declareDef (vArg plus) (`Nat `⇒ `Nat `⇒ `Nat)
  defineFun plus
    (clause (vArg (con (quote zero) [])) ::
      vArg (var "y") ::
      [])
    (var 0 [] ::
      clause (vArg (con (quote suc)
        (vArg (var "x") :: [])) ::
        vArg (var "y") ::
        []))
    (con (quote suc)
      (vArg (def plus
        (vArg (var 1 [] ::
          vArg (var 0 [] :: [])) ::
          [])) ::
      []))
```

Figure 4. Addition, defined by metaprogramming

synonym ``Nat` abbreviates references to the defined name `Nat`, and `vArg` represents the desired visibility and relevance settings of the arguments. Once declared, `plus` is defined using `defineFun`, which takes a name and a list of clauses, defining the function by side effect. Each clause consists of a pattern and a right-hand side. Patterns have their own datatype, while right-hand sides are Terms. The name `con` is overloaded: in patterns, it denotes a pattern that match a particular constructor, while in Terms, it denotes a reference to a constructor.

2.2 Higher Inductive Types

Homotopy type theory [17] is a research program that aims to develop univalent, higher-dimensional type theories. A type theory is *univalent* when equivalences between types are considered equivalent to equalities between types; it is *higher-dimensional* when we allow non-trivial identifications that every structure in the theory must nevertheless respect. Identifications between elements of a type are considered to be at the lowest dimension, while identifications between identifications at dimension n are at dimension $n + 1$. Voevodsky added univalence to type theories as an axiom, asserting new identifications without providing a means to compute with them. While more recent work arranges the computational mechanisms of the type theory such that univalence can be derived, as is done in cubical type theories, we are concerned with modeling concepts from homotopy type theory in existing, mature implementations of type theory, so we follow Univalent Foundations Program [17] in modeling

paths using Martin-Löf's identity type. Higher-dimensional structure can arise from univalence, but it can also be introduced by defining new type formers that introduce not only introduction and elimination principles, but also new non-trivial identifications.

In homotopy type theories, one tends to think of types not as collections of distinct elements, but rather through the metaphor of topological spaces. The individual elements of the type correspond with points in the topological space, and identifications correspond to paths in this space.

While there is not yet a general schematic characterization of a broad class of higher inductive types along the lines of Dybjer's inductive families, it is convenient to syntactically represent the higher inductive types that we know are acceptable as if we had such a syntax. Thus, we sometimes specify a higher inductive type similarly to a traditional inductive type by providing its constructors (*i.e.* its points); we additionally specify the higher-dimensional structure by providing additional constructors for paths. For example, figure 5 describes `Circle`, which is a higher inductive type with one point constructor `base` and one non-trivial path constructor `loop`.

```
data Circle : Set where
  base : Circle
  loop : base ≡ base
```

Figure 5. A specification of a higher inductive type

Agda is a programming language that was originally an implementation of Luo's UTT extended with primitive dependent pattern matching, itself a derivative of the Calculus of Constructions and Martin-Löf's intensional type theory. Agda's type theory has since gained a number of new features, among them the ability to restrict pattern matching to that subset that does not imply Streicher's Axiom K, which is inconsistent with univalence. The convenience of programming in Agda, combined with the ability to avoid axiom K, makes it a good laboratory for experimenting with the idioms and techniques of univalent programming while more practical implementations of univalent type theories are under development.

In Agda, we don't have built-in primitives to support the definition of higher inductive type such as `Circle`. One approach is to use Agda's rewrite rules mechanism to define higher inductive types. In this approach, we define the dependent and non-dependent eliminators of a higher inductive type as parameterized modules inside which we declare the computation rules for points as rewrite rules using `{-# REWRITE , ... #-}` pragma. However, Agda's reflection library do not have interfaces to support introducing new pragmas and defining new modules. Another approach to define higher inductive types is to use Dan Licata's method [11]. According to this method, a higher inductive type is defined using type abstraction inside a module. The module consists of a boiler-plate code segment which defines

cite 1994

Cite Ulf's official website

cite Luo's

Cite Cock

cite Cock, Abel, extended abstract at 2016

and's
ortberg,
r's group

```

module Circle where
  private
    data S* : Set where
      base* : S*

  S : Set
  S = S*

  base : S
  base = base*

  postulate
    loop : base ≡ base*

  recS : {C : Set} →
    (cbase : C) →
    (cloop : cbase ≡ cbase) →
    S → C
  recS cbase cloop base* = cbase

  postulate
    βrecS : {C : Set} →
      (cbase : C) →
      (cloop : cbase ≡ cbase) →
      ap (recS cbase cloop) loop ≡ cloop

  indS : {C : S → Set} →
    (cbase : C base) →
    (cloop : transport C loop cbase ≡ cbase) →
    (circle : S) → C circle
  indS cbase cloop base* = cbase

  postulate
    βindS : {C : S → Set} →
      (cbase : C base) →
      (cloop : transport C loop cbase ≡ cbase) →
      apd (indS {C} cbase cloop) loop ≡ cloop

```

Figure 6. An example of a higher inductive type using Licata's encoding

the higher inductive type using a private base type. Inside the module, the recursion and the induction principles acts on the constructors of the private base type. The abstract type is then exported allowing the reduction rules for point constructors to hold definitionally. For example, `Circle` is defined using Dan Licata's method as follows.

Inside the module `Circle`, the type `S` is defined using a private datatype `S*`. The constructor `base` is defined using `base*` and the path `loop` is given as a postulated propositional equality. The recursion and induction principles are defined by pattern matching on the constructor `base*` of the type `S*`, and thus compute as expected. The clients of `Circle` will not have access to the constructor `base*` of the private

type `S*`, as it is not visible outside the module, which prevents them from writing functions that distinguish between multiple constructors of a higher inductive type that may be identified by additional path constructors. The client's *only* access to the constructor is through the provided elimination rules. The following code gives the non-dependent eliminator (sometimes called the *recursion rule*) `recS`.

`recS` ignores the path argument and simply computes to the appropriate answer for the point constructor. The computational behavior for the path constructor `loop` is postulated using reduction rule `βrecS`. The operator `apPerhaps` we should move the discussion of `ap` earlier, so that we don't need the digression is frequently referred to as `cong`, because it expresses that propositional equality is a congruence. However, when viewed through a homotopy type theory lens, it is often called `ap`, as it describes the action of a function on paths. In a higher inductive type, `ap` should compute new paths from old ones.

$$\text{ap} : \{A B : \text{Set}\} \{x y : A\} \\ (f : A \rightarrow B) (p : x \equiv y) \rightarrow f x \equiv f y$$

The following code gives the dependent eliminator or the induction rule `indS` and its computational rules. The dependent eliminator relies on another operation on identifications, called `transport`, that coerces an inhabitant of a family of types at a particular index into an inhabitant at another index. Outside of homotopy type theory, `transport` is typically called `subst` or `replace`, because it also expresses that substituting equal elements for equal elements is acceptable.

$$\text{transport} : \{A : \text{Set}\} \{x y : A\} \rightarrow \\ (P : A \rightarrow \text{Set}) \rightarrow (p : x \equiv y) \rightarrow P x \rightarrow P y$$

In the postulated computation rule for `indS`, the function `apd` is the dependent version of `ap`: it expresses the action of dependent functions on paths.

$$\text{apd} : \{A : \text{Set}\} \{B : A \rightarrow \text{Set}\} \{x y : A\} \rightarrow \\ (f : (a : A) \rightarrow B a) \rightarrow \\ (p : x \equiv y) \rightarrow (\text{transport } B p (f x) \equiv f y)$$

In the above type, the point $(\text{transport } B p (f x))$ lying in space $B(y)$ can be thought of as an endpoint of a path obtained by lifting the path p from $f(x)$ to a path in the total space $\sum_{(x:A)} B(x) \rightarrow A$.

The next section introduces the necessary automation features by describing the automatic generation of eliminators for a variant on Dybjer's inductive families. Section 4 then generalizes this feature to automate the production of eliminators for higher inductive types using Licata's technique. Section 5 revisits Angiuli et al.'s encoding of Darcs's patch theory [4] and demonstrates that the higher inductive types employed in that paper can be generated succinctly using our library.

3 Code Generation for Inductive Types

An inductive type X is a type that is freely generated by a finite collection of constructors. The constructors of X

This sentence does not mean anything. What is a "space"? No one knows that is what "ap" is made clear?

Check whether this is right - be a variant of Luo's inductive types inst

accept zero or more arguments, and result in an X . The constructors can also take an element of type X itself as an argument, but only *strictly positively*: any occurrences of the type constructor X in the type of an argument to a constructor of X must not be to the left of any arrows.

Type constructors can have a number of *parameters*, which may not vary between the constructors, as well as *indices*, which may.

In Agda, constructors are given a function type. In Agda's reflection library, the constructor data-type of the datatype Definition stores the constructors of an inductive type as a list of Names. The type of a constructor can be retrieved by giving its Name as an input to the `getType` primitive. In the following subsections, we will discuss how to use list of constructors and their types to generate code for the elimination rules of an inductive type.

3.1 Non-dependent Elimimators

In Agda, we define an inductive type using `data` keyword. A definition of an inductive datatype declares its type and specifies its constructors. In general, the definition of an inductive datatype D with constructors $g_1 \dots g_n$ has the following form:

```
data D (x1 : P1) ... (xn : Pn) : Q1 → ... → Qn → Set where
  g1 : {i1 : Q1} ... {in : Qn} → A1 → D x1 ... xn i1 ... in
  ⋮
  gn : {j1 : Q1} ... {jn : Qn} → An → D x1 ... xn j1 ... jn
```

The type of A_i can take one of the three forms: a constant type B , the inductive type D , or a function $B_1 \rightarrow \dots \rightarrow B_n \rightarrow D$ with co-domain D . The parameters $(x_1 : P_1) \dots (x_n : P_n)$ are common to all the constructors, and the type of each constructor implicitly encodes the parameter references. However, the indices are different for each constructor. So, the type of each constructor explicitly exhibits the index declaration.

While inductive datatypes are defined by their constructors, it must also be possible to *eliminate* them. This section describes how to generate a non-dependent recursion principle for an inductive type; section 3.2 generalizes this technique to fully-dependent induction principles.

Based on the generic form of the inductive type given above, we can define the following schematic representation for the non-dependent eliminator.

```
f : {x1 : P1} ... {xn : Pn} → {k1 : Q1} → ... → {kn : Qn} →
  D x1 ... xn k1 ... kn → (C : Set) →
  (c1 : {i1 : Q1} → ... → {in : Qn} → A1* → C) → ...
  (cn : {j1 : Q1} → ... → {jn : Qn} → An* → C) → C
```

The type of A_i^* in constructor c_i depends on the type of A_i in the constructor g_i . If A_i is a constant type B , the automation tool replaces A_i^* with B . For inductive type D , the automation tool replaces A_i^* with type $D \rightarrow C$. For function

```
data Vec (A : Set) : Nat → Set where
  [] : Vec A zero
  _::_ : {n : Nat} →
    (x : A) → (xs : Vec A n) →
    Vec A (suc n)
```

Figure 7. Length-indexed lists

type $(B_1 \rightarrow \dots \rightarrow B_n \rightarrow D)$, the automation tool replaces A_i^* with $(B_1 \rightarrow \dots \rightarrow B_n \rightarrow D) \rightarrow (B_1 \rightarrow \dots \rightarrow B_n \rightarrow C)$. The parameter and the index to type D must be reflected in the type of function f . We can retrieve information about the parameter and the index of D from its type. The constructors refer to the parameter and the index using de Bruijn indices. The general schema for the computation rules corresponding to f and constructors g_1, \dots, g_n is given as follows.

$$\begin{aligned} f(g_1 a_1 \dots a_n) C c_1 \dots c_n &= c_1 \dots \\ &\vdots \\ f(g_n a_1 \dots a_n) C c_1 \dots c_n &= c_n \dots \end{aligned}$$

The value of c_i on the right-hand side of the above computation rules depends on the type of g_i . If g_i takes zero arguments, then the corresponding computation rule for g_i evaluates to c_i . If g_i takes one or more arguments $(g_i a_1 \dots a_n)$, then the computation rule evaluates to $(c_i a_1 \dots a_n)$. If g_i takes the inductive type D as an argument $(g_i \dots d_i \dots)$, then the computation rule evaluates to $(c_i \dots (f \dots) \dots)$, encoding the recursor f to evaluate the value for the substructure d_i of D . If g_i takes a function with co-domain D as an argument such as $(g_i \dots d_i^* \dots)$, we need a composite function $(f \circ d_i^*)$ to evaluate the value for the substructure d_i^* . The automation tool handles composites by using a lambda function $(\lambda a \dots \rightarrow (f \dots (d_i^* a \dots) \dots))$.

The Agda datatype `Vec` represents lists of a known length. It is defined in figure 7. For `Vec`, the recursion principle says that to define a mapping $f : \text{Vec } A \ n \rightarrow C$, it suffices to define the action of f on inputs `[]` and `_::_`.

To define the action of f on inputs `[]` and `_::_`, we need elements c_1 and c_2 of the following type.

```
c1 : C
c2 : {m : Nat} → (x : A) → (xs : Vec A m) →
  C → C
```

The recursor f maps the constructor `[]`, which takes zero arguments, to c_1 . The constructor `_::_` takes a constant argument x and an argument xs of the inductive type `Vec`. f maps $(x :: xs)$ to $(c_2 \ x \ xs \ (f \ xs \ C \ c_1 \ c_2))$. To construct the type of the recursion rule for `Vec`, we need to build the type of c_1 and c_2 . Since `[]` is not a function type, we can map it directly to $c_1 : C$. We can retrieve the static type information of `_::_` using reflection primitives, and use that to construct the type of c_2 . The constructor `pi` of type `Term` encodes the abstract syntax tree (AST) representation of `_::_` (fig. 8). We can retrieve and traverse the AST of `_::_`,

The usual
tion inste
fies the p
over the c
eliminato
then the
arately ov
motive, in
ing the in
necessary
method, a
quantifies
more for
of elimina

```

pattern _[_v]⇒_ a s b = pi (vArg a) (abs s b)
pattern _[_h]⇒_ a s b = pi (hArg a) (abs s b)

(agda-sort (lit 0) [ "A" h]⇒          -- A
(def (quote Nat) [] [ "n" h]⇒        -- n
  (var 1 [] [ "x" v]⇒                -- x
    (def (quote Vec)                  -- xs : Vec A n
      (vArg (var 2 []) ::
        vArg (var 1 []) :: [])
      [ "xs" v]⇒
    def (quote Vec)                    -- Vec A (suc n)
      (vArg (var 3 []) ::
        vArg (con (quote suc)
          (vArg (var 2 []) :: []))
        :: []))))))

```

Figure 8. Abstract syntax tree for constructor $_{::}$

```

(agda-sort (lit 0) [ "A" h]⇒          -- A
(def (quote Nat) [] [ "n" h]⇒        -- n
  (def (quote Vec) (vArg (var 1 []) :: -- Vec A n
    vArg (var 0 []) :: []) [ "_" v]⇒
  (agda-sort (lit 0) [ "C" v]⇒        -- C
    (var 0 [] [ "_" v]⇒                -- c1
      ((def (quote Nat) [] [ "m" h]⇒   -- c2
        (var 5 [] [ "x" v]⇒            -- x
          (def (quote Vec) (vArg (var 6 []) ::
            vArg (var 1 []) :: []) [ "xs" v]⇒
            (var 4 [] [ "_" v]⇒ var 5 [])))) -- C → C
        [ "_" v]⇒ var 2 [])))))) -- C

```

Figure 9. Abstract syntax tree for recursor f on Vec

and add new type information into it to build a new type representing $c2$.

During the traversal of abstract syntax tree of the type of $_{::}$, when the type Vec occurs directly in a non-codomain position, we add the type C next to it. For example, in figure 8, a new function is built from the argument $(xs : Vec A n)$ by modifying it to $(Vec A n) \rightarrow C$ (fig. 9). Constant types require no modifications. Therefore, we copy $(x : A)$ into the new type without any changes. Finally, we change the codomain $Vec A (suc n)$ of $_{::}$ to C resulting in an abstract syntax tree representation of the type $c2$. We repeat this process for each of the constructors.

The variable $c2$ refers to the parameter and the index using de Bruijn indices. During the construction of the output type $c2$, the automation tool updates the de Bruijn indices accordingly. The constructor `data-type` contains the count cp of parameters occurring in a defined type. It also encodes the constructors of the type as a list of Names. Metaprograms can retrieve the index count by finding the difference between cp and the length of the constructor list. Some constructors might not take the same number of indices as the parent type. For example, in the case of Vec , the constructor `[]` excludes the index Nat from its type. We do not have any reflection primitive to retrieve the index count from a constructor name. A workaround is to pass the index count of each constructor explicitly to the automation tool.

Once we have the AST of $c2$, we can build the type of the recursion rule f for Vec (fig.9). To encode the mapping Vec

```

(cclause
  (vArg (con (quote _[_v]⇒)
    (vArg (var "x") ::
      vArg (var "xs") :: []))
    vArg (var "C") ::
    vArg (var "c1") ::
    vArg (var "c2") :: [])
  (var 0
    (vArg (var 4 []) ::
      vArg (var 3 []) ::
      vArg (def f
        (vArg (var 3 [])
          vArg (var 2 []) ::
          vArg (var 1 []) ::
          vArg (var 0 []) :: []))))))

```

Figure 10. Clause definition for the computation rule of $_{::}$

```

data W (A : Set) (B : A → Set) : Set where
  sup : (a : A) → (B a → W A B) → W A B

```

Figure 11. W-Type

$A n \rightarrow C$ in the recursion type, we need to declare C . We can use the constructor `agda-sort` to introduce the type $(C : Set)$. The type of the recursion rule f is given as follows.

```

f : {A : Set} → {n : Nat} → Vec A n →
  (C : Set) →
  (c1 : C) →
  (c2 : {m : Nat} → (x : A) →
    (xs : Vec A m) → C → C) →
  C

```

The above type is declared using `declareDef`. We can build the computation rule representing the action of function f on `[]` and $_{::}$ using `clause` (fig. 10). The first argument to `clause` encodes variables corresponding to the above type, and it also includes the abstract representation of $_{::}$ on which the pattern matching should occur. The second argument to `clause`, which is of type `Term`, refers to the variables in the first argument using de Bruijn indices, and it encodes the output of the action of function f on $_{::}$. The constructor `var` in `Pattern` is used to introduce new variables in the clause definition. The type `Pattern` also has another constructor `con` used to represent the pattern matching term. The type `Term` has similar constructors `var` and `con`, but with different types, used to encode the output of the recursion rule. The computation rules corresponding to the above type is given as follows.

```

f [] C c1 c2 = c1
f (x :: xs) C c1 c2 = c2 x xs (f xs C c1 c2)

```

A clause definition, which evaluates to the above computation rule of Vec pattern matching on $_{::}$, is given in figure 10. The de Bruijn index reference increments right to left starting from the last argument. The above clause definition is defined using `defineFun` primitive, and the function f is brought into scope by `unquotedec1`.

Lets consider another example, the W-type, to review the automation process for the recursion principle. W-type has a constructor sup given by the following type.

$$\text{sup} : (a : A) \rightarrow (B \ a \rightarrow W \ A \ B) \rightarrow W \ A \ B$$

To define the action of a function $f : W \ A \ B \rightarrow C$ on input sup, we need a function d of the following type.

$$d : (a : A) \rightarrow (B \ a \rightarrow W \ A \ B) \rightarrow (B \ a \rightarrow C) \rightarrow C$$

The type of d is built by traversing the AST of sup. During the traversal of the AST of sup, the first argument to sup, which is a constant type A, is copied directly into the AST of d. The second argument $(B \ a \rightarrow W \ A \ B)$, which is a function with co-domain $W \ A \ B$, is modified to $(B \ a \rightarrow W \ A \ B) \rightarrow (B \ a \rightarrow C)$. Finally, the co-domain $W \ A \ B$ of sup is replaced by C. The computation rule corresponding to sup is given as follows.

$$f \ (\text{sup} \ a \ b) \ C \ d = d \ a \ b \ (\lambda \ v \rightarrow f \ (b \ v) \ C \ d)$$

In the above computation rule, the third argument to d is a composition of functions f and b. The automation tool composes functions inside a lambda directly, using the Term constructor lam. The arguments to lam are referenced using de Bruijn indices inside the lambda body. So, the de Bruijn indices for referring variables outside the lambda body are updated accordingly. For example, inside the lambda body, the reference 0 refers to the lambda argument v, and the index references to the variables outside the lambda body start from 1 and increment towards the left.

In the automation tool, generateRec interface is used to generate the recursion rule f. The implementation of generateRec is given as follows.

```
generateRec : Arg Name → Name →
  (indexList : List Nat) → TC T
generateRec (arg i f) t indLs =
  do indLs' ← getIndex t indLs
  cns ← getConstructors t
  lcons ← getLength cns
  cls ← getClause lcons zero t f indLs cns
  RTy ← getType t
  funType ← getRType t indLs' zero RTy
  declareDef (arg i f) funType
  defineFun f cls
```

generateRec uses getClause and getRType to build the computation and elimination rules respectively. It takes three arguments: the name of the function to be defined (represented by an element of type Arg Name), the quoted Name of the type and a list containing the index count of the individual constructors. generateRec can be used to automate the generation of recursion rules for inductive types having the general schema given at the beginning of this section. The recursion rule generated by generateRec is brought into scope using unquoteDecl as follows.

```
unquoteDecl f = generateRec (vArg f)
  (quote Vec)
  (0 :: 1 :: [])
```

The third argument to generateRec is a list consisting of the index count for the constructors. It is required to pass the index count for each constructor explicitly as the Agda reflection library does not have built-in primitives to retrieve the index value.

3.2 Dependent Eliminators

The dependent eliminator for a datatype, also known as the *induction principle*, is used to eliminate elements of a datatype when the type resulting from the elimination mentions the very element being eliminated. We can define the general schema for the induction principle as follows.

$$\begin{aligned} f : \{x_1 : P_1\} \dots \{x_n : P_n\} &\rightarrow \{k_1 : Q_1\} \rightarrow \dots \rightarrow \{k_n : Q_n\} \rightarrow \\ (d : D \ x_1 \dots x_n \ k_1 \dots k_n) &\rightarrow (C : D \rightarrow Set) \rightarrow \\ (c_1 : \{i_1 : Q_1\} \rightarrow \dots \rightarrow \{i_n : Q_n\} &\rightarrow A_i^* \rightarrow C(g_1 \dots)) \rightarrow \dots \\ (c_n : \{j_1 : Q_1\} \rightarrow \dots \rightarrow \{j_n : Q_n\} &\rightarrow A_n^* \rightarrow C(g_n \dots)) \rightarrow C \ d \end{aligned}$$

Similar to the recursion principle, the type of A_i^* in constructor c_i depends on the type of A_i in the constructor g_i . If A_i is a constant type B, the automation tool replaces A_i^* with B. For inductive type D, the automation tool replaces A_i^* with type $(d_1 : D) \rightarrow C \ d_1$. For function type $(B_1 \rightarrow \dots \rightarrow B_n \rightarrow D)$, the automation tool replaces A_i^* with $(y : B_1 \rightarrow \dots \rightarrow B_n \rightarrow D) \rightarrow ((b_1 : B_1) \rightarrow \dots \rightarrow (b_n : B_n) \rightarrow C(y \ b_1 \dots b_n))$. The computation rules corresponding to the induction principle follows the same pattern as for the recursion rule.

For the inductive type W, the induction principle says that to define a mapping $f : (w : W \ A \ B) \rightarrow C \ w$, it suffices to define the action of f on the constructor sup. To define the action of f on sup, we need an element d of the following type.

$$\begin{aligned} d : (a : A) &\rightarrow (y : B \ a \rightarrow W \ A \ B) \rightarrow \\ (z : (v : B \ a) &\rightarrow C \ (y \ v)) \rightarrow \\ C \ (\text{sup} \ a \ y) \end{aligned}$$

To build the dependent eliminator, we need the type of the function d. We can construct the AST of d using the static type information obtained from sup. To construct d, during the traversal of the AST of sup, we copy the constant $(a : A)$ directly without any changes as in the case of the non-dependant eliminator. When we identify a function $y : B \ a \rightarrow W \ A \ B$ with codomain $W \ A \ B$, we add a new function $(z : (v : B \ a) \rightarrow C \ (y \ v))$ with the same arguments as in y and codomain $C \ (y \ v)$, which depends on the action of y on v. Finally, the co-domain $W \ A \ B$ of the constructor sup is changed to $C \ (\text{sup} \ a \ y)$, which depends on the action of the constructor sup on inputs a and y (fig. 12).

We can construct the type of the induction principle f using d. The type C in the mapping f depends on the element of the input type $W \ A \ B$. The type of the induction principle f (fig. 12) is given as follows.

```

(agda-sort (lit 0) [ "A" h]⇒                -- A
((var 0 [] [ "_" v]⇒                        -- B : A → Set
  agda-sort (lit 0)) [ "B" h]⇒
  (def (quote W) (vArg (var 1 []) ::
    vArg (var 0 []) :: []) [ "C" v]⇒
    ((def (quote W) (vArg (var 2 []) ::
      vArg (var 1 []) :: []) [ "_" v]⇒
      agda-sort (lit 0)) [ "C" v]⇒
      ((var 3 [] [ "a" v]⇒                -- a : A
        ((var 3 (vArg (var 0 []) :: [])
          [ "_" v]⇒
            def (quote W) (vArg (var 5 []) ::
              vArg (var 4 []) :: []) [ "y" v]⇒
              ((var 4 (vArg (var 1 []) :: [])
                [ "v" v]⇒
                  var 3 (vArg (var 1 (vArg (var 0 []))
                    :: [])) :: [])
                  [ "z" v]⇒
                    var 3                -- C (sup a y)
                  (vArg
                    (con (quote sup)
                      (vArg (var 2 []) ::
                        vArg (var 1 []) :: []))
                    :: [])))
                  [ "_" v]⇒ var 1 (vArg (var 2 []) :: [])))))) -- C c

```

Figure 12. Abstract syntax tree for the dependent eliminator of W

$$\begin{aligned}
 f : \{A : \text{Set}\} \{B : A \rightarrow \text{Set}\} \rightarrow \\
 (c : W \ A \ B) \rightarrow (C : W \ A \ B \rightarrow \text{Set}) \rightarrow \\
 ((a : A) \rightarrow (y : B \ a \rightarrow W \ A \ B) \rightarrow \\
 (z : (v : B \ a) \rightarrow C \ (y \ v)) \rightarrow \\
 C \ (\text{sup } x \ y)) \rightarrow C \ c
 \end{aligned}$$

The computation rule corresponding to the above type is the same as the computation rule of the recursion principle of W . It is constructed using clause definitions following the same approach as the recursion principle. We can automate the generation of the induction rule f and its corresponding computation rules using `generateInd` interface. The following code gives the implementation of `generateInd`.

```

generateInd : Arg Name → Name →
  (indexList : List Nat) → TC T
generateInd (arg i f) t indLs =
  do id' ← getIndex t indLs
     cns ← getConstructors t
     lcns ← getLength cns
     cls ← getClauseDep lcns zero t f id' cns
     RTy ← getType t
     funType ← getRtypeInd t zero id' RTy
     declareDef (arg i f) funType
     defineFun f cls

```

`generateInd` uses `getClauseDep` to generate the clause definitions representing the computation rules. The abstract representation of the type is provided by `getRtypeInd`. The function f generated by `generateInd` is brought into scope by `unquoteDecl` as follows.

```

unquoteDecl f = generateInd (vArg f)
  (quote W) []

```

An empty list is passed to `generateInd` as W has no index. An empty list can also be passed if all the constructors of a type has the same number of index as the parent type. But if any one constructor has an index count different from the index count of the parent type, then the index count of all the constructors should be passed explicitly.

4 Code Generation for Higher Inductive Types

In Agda, there are no built-in primitives to support the definition of higher inductive types. However, we can still define a higher inductive type with a base type using Dan Licata's [11] method as discussed in section 2.1. In this section, we discuss the automation of code generation for the boilerplate code segments defining the higher inductive type. We also describe how to automate the code generation for the elimination and the computation rules of the higher inductive type using static type information obtained from the base type.

4.1 Higher Inductive Type Definition

In Agda, we define an inductive type using the `data` keyword. A datatype declaration characterizes a datatype by declaring its type and specifying its constructors. Consider the following generic form for the definition of an inductive type W with constructors $g_1 \dots g_n$.

$$\begin{aligned}
 \text{data } W^* \ (x_1 : P_1) \dots (x_n : P_n) : Q_1 \rightarrow \dots \rightarrow Q_n \rightarrow \text{Set } \ell \text{ where} \\
 g_1^* : \{i_1 : Q_1\} \dots \{i_n : Q_n\} \rightarrow \text{Type}_1 \rightarrow W^* x_1 \dots x_n i_1 \dots i_n \\
 \vdots \\
 g_n^* : \{j_1 : Q_1\} \dots \{j_n : Q_n\} \rightarrow \text{Type}_n \rightarrow W^* x_1 \dots x_n j_1 \dots j_n
 \end{aligned}$$

The parameters $(x_1 : P_1) \dots (x_n : P_n)$ are common to all the constructors, and the type of each constructor implicitly encodes the parameter references. However, the indices may be different for each constructor. So, the type of each constructor explicitly exhibits the index declaration. During the construction of a higher inductive type, we have to represent the parameters and the indices as explicit arguments in the types of the constructors. We define a higher inductive type W as a top-level definition using a base type W^* similar to the module `Circle` in section 2.1. The reflection type `Definition` provides us the type and the constructors of the base type W^* . We copy the type of W^* to W and for the constructors $g_1 \dots g_n$ of W , we traverse the abstract representation of the type of $g_1^* \dots g_n^*$ respectively replacing the occurrence of W^* to W in every strictly positive position. Consider a constructor g_i^* that has the following type.

$$g_i^* : (A \rightarrow W^*) \rightarrow (B \rightarrow W^*) \rightarrow C \rightarrow W^* \rightarrow W^*$$

We built the type of g_i by traversing the abstract syntax tree of g_i^* and replacing the base type W^* with the higher inductive type W . The abstract syntax tree of g_i^* incorporates

This looks like Dybjer's Luo's pre of datatype Agda's...

Totally unnecessary to give keyword who can here either this or call out.

Use a better metavariable

Why not constructors?

This more schematic notation shows much earlier

What does "hibits" mean?

These series are mixing description a datatype technique coding H Agda metagrams. To apart!

Not "we"

the type of the parameters and the indices if present. We have to retain the parameters and the indices explicitly during the construction of g_i . The following equation represents the type of the constructor g_i .

$$g_i : (A \rightarrow W) \rightarrow (B \rightarrow W) \rightarrow C \rightarrow W \rightarrow W$$

We explicitly pass the types of the path constructors to the automation tool. The higher inductive type definition of `Circle` in section 2.1 represents the path constructors as propositional equalities. The automation tool takes the path types as input and declares them as propositional equalities using the reflection primitive `declarePostulate`. We introduce a new data type `ArgPath` to input the path types to the automation tool.

*data ArgPath {ℓ₁} : Set (Isuc ℓ₁) where
argPath : Set ℓ₁ → ArgPath*

The constructor `argPath` takes the type of a path constructor as input. We define the generic form of a higher inductive type as follows.

data-hit (quote W) W
Wpoints (g₁ :: ... :: g_n :: [])
Wpaths (p₁ :: ... :: p_n :: [])
(argPath
{x₁ : P₁} → ... → {x_n : P_n} →
{i₁ : Q₁} → ... → {i_n : Q_n} → Type_{e₁} →
(g_i{x₁}...{x_n}{i₁}...{i_n}...) ::
:
argPath
{x₁ : P₁} → ... → {x_n : P_n} →
{j₁ : Q₁} → ... → {j_n : Q_n} → Type_{e_n} →
(g_i{x₁}...{x_n}{j₁}...{j_n}...) :: [])*

We define holders `Wpoints` for point constructors and `Wpaths` for path constructors as part of the higher inductive type definition of W . We cannot retrieve the constructors of the higher inductive type W using `Definition`. Therefore, `Wpoints` and `Wpaths` act as the only references for the constructors of W . The elements of the `argPath` list represent the type of the path constructors $p_1 \dots p_n$ respectively. We explicitly include the parameter references $\{x_1 : P_1\} \dots \{x_n : P_n\}$ and the index references $\{i_1 : Q_1\} \dots \{i_n : Q_n\}$ in the type of the arguments to `argPath`. The points $g_1 \dots g_n$ are not in scope when used in the identity type passed to `argPath`. The automation tool uses the base type constructors $g_1^* \dots g_n^*$ as dummy arguments in the place of $g_1 \dots g_n$ respectively. The automation tool implements the interface `data-hit` as follows.

*data-hit : ∀{ℓ₁} (baseType : Name) → (indType : Name) →
(pointHolder : Name) → (lcons : List Name) →
(pathHolder : Name) → (lpaths : List Name) →
(lpathTypes : (List (ArgPath {ℓ₁}))) → TC T
do defineHindType baseType indType
lcons' ← getConstructors baseType*

```
defineHitCons baseType indType lcons' lcons
lpathTypes' ← getPathTypes baseType indType lcons' lcons lpathTypes
defineHitPathCons lpaths lpathTypes'
definePointHolder pointHolder lcons
definePathHolder pathHolder lpaths
```

The higher inductive type W , the points $g_1 \dots g_n$, the paths $p_1 \dots p_n$, and the holders `Wpoints` and `Wpaths` are brought into scope by `unquoteDecl`. In the above implementation of `data-hit`, `defineHindType` defines the higher inductive type as a top-level definition using the base type. `defineHitCons` specifies the point constructors of the higher inductive type using the type information obtained from the constructors of the base type, and `defineHitPathCons` builds the paths constructors of the higher inductive type using the `argPath` list. The following code automates the generation of the higher inductive type definition for `Circle` given in section 2.1.

```
unquoteDecl S Spoints base Spaths loop =
  data-hit (quote S*) S
  Spoints (base :: []) -- point constructors
  Spaths (loop :: []) -- path constructors
  (argPath (base* ≡ base*) :: []) -- base replaces base*
```

The identity type input (`base* ≡ base*`) to `argPath` represents the type of the path loop, and it uses the inductive type constructor `base*` as a dummy argument in the place of the higher inductive type constructor `base`. The constructor `base` comes into scope only during the execution of `unquoteDecl`, and so cannot be used in the identity type reference in `argPath`. We use the constructor `base*` of type `S*` as dummy argument because the type of `base*` is similar to `base`, and has the same references for the common arguments. The automation tool traverses the abstract syntax tree of `loop` and replaces the occurrences of `base*` with `base`.

4.2 Non-dependent Eliminator

Non-dependent eliminator or the recursion principle of a higher inductive type W maps the points and paths of W to an output type P . Let $g_1 : A \rightarrow W$ and $g_2 : B \rightarrow W$ be the point constructors, and $l : (\alpha : A) \rightarrow (\beta : B) \rightarrow (g_1 \alpha) \equiv (g_2 \beta)$ be the path constructor of W . To define a mapping $f : W \rightarrow P$, we need points $d_1 : A \rightarrow P$ and $d_2 : B \rightarrow P$ and a path $h : (\alpha : A) \rightarrow (\beta : B) \rightarrow (d_1 \alpha) \equiv (d_2 \beta)$ in the space P . The path constructor l can take arguments of any type including the higher inductive type W but for simplicity we will consider only arguments of constant type.

To construct the recursion principle f , we need to build the type of points d_1, d_2 and path h . The type of d_1 and d_2 are built from the abstract syntax tree of points g_1 and g_2 respectively using the approach described in section (3.1). We build the type of h by traversing the abstract syntax tree of l . During the traversal, we replace the points g_1 and g_2 , which are arguments to the identity type, in the co-domain of the path l by the points d_1 and d_2 respectively. In figure

```

(pi (vArg (agda-sort (lit 0)))           -- P
  (abs "P"
    (pi (vArg (pi (vArg (def (quote A) [])) -- d1 : A → P
      (abs "_" (var 1 []))))             -- P ref
      (abs "d1"
        (pi (vArg (pi (vArg (def (quote B) [])) -- d2 : B → P
          (abs "_" (var 2 []))))           -- P ref
          (abs "d2"
            (pi (vArg
              (pi (vArg (def (quote A) [])) -- h
                (abs "α"
                  (pi (vArg (def (quote B) [])) -- β : B
                    (abs "β"
                      (def (quote _≡_)
                        (vArg (var 3) -- d1 ref
                          (vArg (var 1 [] :: []) :: -- α ref
                            (vArg (var 2) -- d2 ref
                              (vArg (var 0 [] :: []) :: [])))))) -- β ref
                        (abs "h"
                          (pi (vArg (def (quote W) [])) -- W → P
                            (abs "_" (var 4 []))))))))))))

```

Figure 13. Abstract syntax tree for constructor f

(13), the arguments of the identity type in the abstract syntax tree of h references the points d_1 and d_2 using de Bruijn index 3 and 2 respectively. We copy the other two arguments of l , which are of constant type A and B , without any changes into the type of h .

The recursion rule f corresponding to figure (13) is given by the following equation.

$$\begin{aligned}
 f : (P : \text{Set}) &\rightarrow (d_1 : A \rightarrow P) \rightarrow (d_2 : B \rightarrow P) \rightarrow \\
 (h : (\alpha : A) &\rightarrow (\beta : B) \rightarrow (d_1 \alpha \equiv (d_2 \beta))) \rightarrow \\
 W &\rightarrow P
 \end{aligned}$$

The computation rules for constructors g_1 and g_2 give the action of eliminator f on input $g_1(\alpha)$ and $g_2(\beta)$ respectively. The computation rules say that the function f maps the points $g_1(\alpha)$ and $g_2(\beta)$ to the points $d_1(\alpha)$ and $d_2(\beta)$ respectively in the output type P . We specify the clause definition to build the computation rules for g_1 and g_2 in the same way as for a regular inductive type (sec. 3.1) except that the clause definition for the higher inductive type W has an additional reference variable to the path h . The following definitional equalities give the computation rules for g_1 and g_2 .

$$\begin{aligned}
 f(P, d_1, d_2, h, g_1(\alpha)) &\equiv d_1(\alpha) \\
 f(P, d_1, d_2, h, g_2(\beta)) &\equiv d_2(\beta)
 \end{aligned}$$

The computation rule for the path constructor l defines the action of function f on the path $(l \alpha \beta)$. In type theory, the functions are functorial, and they preserve the path structures in their mapping to the output type [17]. The function $ap_f(??)$ defines the action of the non-dependent eliminator f on the path l . Using $(??)$, we can build the computation

```

(def (quote _≡_)
  (vArg (def (quote ap)
    (vArg (lam visible (abs "x"
      (def (quote f)
        (vArg (var 6 [] :: -- P ref
          vArg (var 5 [] :: -- d1 ref
            vArg (var 4 [] :: -- d2 ref
              vArg (var 3 [] :: -- h ref
                vArg (var 0 [] :: []))))) :: -- x ref
        vArg (def (quote l)
          (vArg (var 1 [] :: -- α ref
            vArg (var 0 [] :: [])) :: [])) :: -- β ref
          vArg (var 2
            (vArg (var 1 [] :: -- α ref
              vArg (var 0 [] :: [])) :: []))))) :: -- h ref
        vArg (var 0 [] :: [])) :: [])) :: -- β ref

```

Figure 14. Abstract syntax tree for the term representing the action of function f on path l

rule for the path l as follows.

$$\begin{aligned}
 \beta_f : (P : \text{Set}) &\rightarrow (d_1 : A \rightarrow P) \rightarrow (d_2 : B \rightarrow P) \rightarrow \\
 (h : (\alpha : A) &\rightarrow (\beta : B) \rightarrow (d_1 \alpha \equiv (d_2 \beta))) \rightarrow \\
 \{\alpha : A\} &\rightarrow \{\beta : B\} \rightarrow \\
 ap(\lambda x \rightarrow f(P, d_1, d_2, h, x)) &(l \alpha \beta) \equiv (h \alpha \beta)
 \end{aligned}$$

The computation rule β_f exists only as propositional equality. We build the type of β_f using the same approach as for the recursion rule f . The type of f and β_f is identical except for the mapping $W \rightarrow P$ in f which is replaced by the term representing the action of function f on the path $(l \alpha \beta)$. The function ap applies f , which is nested inside a lambda function, on the arguments $(g_1 \alpha)$ and $(g_2 \beta)$ of the path $(l \alpha \beta)$. We use the constructor `lam` of `Term` to introduce the lambda function with argument x (fig. 14). Inside the lambda body, the argument x gets a de Bruijn index reference of zero and we update the remaining de Bruijn indices accordingly. The application of function f to the path $(l \alpha \beta)$ substitutes the points $(g_1 \alpha)$ and $(g_2 \beta)$ for the lambda argument x , and it evaluates to the path $(h \alpha \beta)$ in the output type P . We implement the `generateRecHit` interface as follows.

```

1 generateRecHit : Arg Name → List (Arg Name) →
2   (baseType : Name) → (indexList : List Nat) →
3   (baseRec : Name) → (indType : Name) →
4   (points : List Name) → (paths : List Name) → TC T
5 generateRecHit (arg i f) argD baseType indLs baseRec indType points paths =
6   do lcons ← getConstructors baseType
7     lpoints ← getLength points
8     lpaths ← getLength paths
9     clauses ← getPathClause lpoints lpaths baseRec
10    RTy ← getType baseType
11    funTypePath ← getRTypePath baseType indType baseRec indLs paths zero RTy
12    declareDef (arg i f) funTypePath
13    defineFun f clauses
14    generateβRecHit argD baseType indLs baseRec indType f points paths

```

`generateRecHit` takes the base type recursion rule as input and uses that to map the points $g_1(\alpha)$ and $g_2(\beta)$ to $d_1(\alpha)$ and $d_2(\beta)$ respectively in the abstract syntax tree of the path $(l \alpha \beta)$. The second argument `argD` is a list of terms

representing the computation rules for the path constructors. The `generateβRecHit` interface takes `argD` as input and builds the computation rule for the path constructor l . Other inputs to `generateRecHit` are the point and path holders declared during the higher inductive type definition of W . The following automation code generates the recursion rule and the computation rules for `Circle`.

```
unquoteDecl recS1* = generateRec (vArg recS1*)
  (quote S1*) (0 :: [])

unquoteDecl recS1 βrecS1 = generateRecHit (vArg recS1)
  ((vArg βrecS1) :: [])
  (quote S1*) (0 :: [])
  (quote recS1*)
  (quote S1) S1points S1paths
```

The term βrecS_1 represents the computation rule for the path constructor loop and is brought into scope by `unquoteDecl`.

4.3 Dependent Eliminator

Dependent eliminator or the induction principle of a higher inductive type W is a dependent function that maps an element ω of W to an output type $P \omega$. For the type W with point constructors $g_1 : A \rightarrow W$ and $g_2 : B \rightarrow W$ and path constructor $l : (\alpha \rightarrow A) \rightarrow (\beta \rightarrow B) \rightarrow (g_1 \alpha) \equiv (g_2 \beta)$, to define a mapping $f' : (\omega : W) \rightarrow P \omega$, we need $d_1 : (\alpha : A) \rightarrow P \alpha$, $d_2 : (\beta : B) \rightarrow P \beta$, and $h : (\alpha \rightarrow A) \rightarrow (\beta : B) \rightarrow \text{transport } P (l \alpha \beta) (d_1 \alpha) \equiv (d_2 \beta)$, where $(h \alpha \beta)$ is a heterogeneous path transported over $(l \alpha \beta)$ lying in the space $P \beta$. `Equation(??)` gives the type of `transport`.

In the abstract syntax tree of f' (fig. 15), the type of constructors d_1 and d_2 are built from the type of g_1 and g_2 respectively using the same approach as in section (3.2). When building the type of d_1 , we copy the constant argument A directly into the abstract syntax tree of d_1 , and for the codomain P , which depends on the action of constructor g_1 on the constant argument $\alpha : A$, we pass the de Bruijn index reference of α as an argument to the constructor g_1 built using the reflection construct `con`. Similarly, for d_2 , we copy the type B directly into the abstract syntax tree of d_2 and pass the de Bruijn index of $\beta : B$ to the constructor g_2 in the co-domain. We build the type of path h by traversing the abstract syntax tree of l and adding relevant type information into it. We copy the constant arguments of type A and B without modification. For the codomain of l , which is the identity type $(g_1 \alpha) \equiv (g_2 \beta)$, we insert the quoted name of function `transport` with arguments P , the path $(l \alpha \beta)$ and $(d_1 \alpha)$. We apply the base eliminator to the arguments of the path $(l \alpha \beta)$ to form the elements $(d_1 \alpha)$ and $(d_2 \beta)$. The following declaration gives the type of f' corresponding to

```
(pi (vArg (`W => agda-sort (lit 0)))
  (abs "P"
    (pi (vArg
      (pi (vArg `A)
        (abs "α"
          (var 1 (vArg (con (quote g1)
            (vArg (var 0 []) :: [])) :: []))))))
      (abs "d1"
        (pi (vArg
          (pi (vArg `B)
            (abs "β"
              (var 2 (vArg (con (quote g2)
                (vArg (var 0 []) :: [])) :: []))))))
              (abs "d2"
                (pi (vArg
                  (pi (vArg `A)
                    (abs "α"
                      (pi (vArg `B)
                        (abs "β"
                          (def (quote _≡_)
                            (vArg (def (quote transport)
                              (vArg (var 4 []) ::
                                vArg (def (quote l) (vArg (var 1 []) ::
                                  vArg (var 0 []) :: [])) ::
                                  vArg (var 3 (vArg (var 1 []) :: []) :: []) ::
                                  vArg (var 2 (vArg (var 0 []) :: []) :: [])))))))))
                              (abs "h"
                                (pi (vArg `W)
                                  (abs "ω" (var 4 (vArg (var 0 []) :: []))))))))))))))
```

Figure 15. Abstract syntax tree of function f'

figure (15).

$$\begin{aligned}
 f' : (P : W \rightarrow \text{Set}) \rightarrow \\
 (d_1 : (\alpha : A) \rightarrow P(g_1 \alpha)) \rightarrow (d_2 : (\beta : B) \rightarrow P(g_2 \beta)) \rightarrow \\
 (h : (\alpha : A) \rightarrow (\beta : B) \rightarrow \\
 \text{transport } P (l \alpha \beta) (d_1 \alpha) \equiv (d_2 \beta)) \rightarrow \\
 (\omega : W) \rightarrow P(\omega)
 \end{aligned}$$

We build the computation rules corresponding to the mapping f' for the point constructors g_1 and g_2 using the same approach as for the non-dependent eliminator f . The following equations give the computation rules for the point constructors g_1 and g_2 .

$$\begin{aligned}
 f'(P, d_1, d_2, h, g_1(\alpha)) &\equiv d_1(\alpha) \\
 f'(P, d_1, d_2, h, g_2(\beta)) &\equiv d_2(\beta)
 \end{aligned}$$

The computation rule $\beta_{f'}$ for the heterogeneous path h depends on `apd (??)` which gives the action of dependent function f' on the path h . We build the type of $\beta_{f'}$ using `(??)` as follows.

$$\begin{aligned}
 \beta_{f'} : (P : W \rightarrow \text{Set}) \rightarrow \\
 (d_1 : (\alpha : A) \rightarrow P(g_1 \alpha)) \rightarrow (d_2 : (\beta : B) \rightarrow P(g_2 \beta)) \rightarrow \\
 (h : (\alpha : A) \rightarrow (\beta : B) \rightarrow \\
 \text{transport } P (l \alpha \beta) (d_1 \alpha) \equiv (d_2 \beta)) \rightarrow \\
 \text{apd}_{(\lambda x \rightarrow f'(P, d_1, d_2, h, x))} (l \alpha \beta) \equiv (h \alpha \beta)
 \end{aligned}$$




Figure 16. Abstract syntax tree for the term representing the action of dependent function f' on path l

We build the abstract syntax tree for the computation rule $\beta_{f'}$ in the same way as for induction principle f' . In the codomain of $\beta_{f'}$ (fig. 16), which is an identity type, we insert the abstract representation of the function apd with arguments f' , nested inside a lambda body, and the path $(l \alpha \beta)$. We pass the path $(h \alpha \beta)$, built using de Bruijn indices, as the second argument to the identity type. The following code gives the implementation of the `generateIndHit` interface in the automation tool.

```
generateIndHit : Arg Name → List (Arg Name) →
  (baseType : Name) → (indLs : List Nat) →
  (baseElm : Name) → (indType : Name) →
  (points : List Name) → (paths : List Name) → TC T
generateIndHit (arg i f) argD baseType indLs baseRec indType points paths =
  do indLs' ← getIndex baseType indLs
  lcons ← getConstructors baseType
  lpoints ← getLength points
  lpaths ← getLength paths
  clauses ← getPathClauseDep lpoints lpaths baseType baseRec indLs' lcons
  RTy ← getType baseType
  funTypePath ← getRTypePathDep baseType indType baseRec points paths zero indLs' RTy
  declareDef (arg i f) funTypePath
  defineFun f clauses
  generateβIndHit argD baseType indLs baseRec indType f points paths
```

The interface `generateIndHit` takes the base eliminator as an argument and uses that to map the point constructors g_1 and g_2 to output d_1 and d_2 when building the codomain of the path constructor h . `generateβIndHit` builds the computation rule for the path constructor h . The following code builds the induction principle and the computation rules for `Circle`.

```
unquoteDecl indS1* = generateInd (vArg indS1*)
  (quote S1*) []
unquoteDecl indS1 βindS1 = generateIndHit (vArg indS1)
  ((vArg βindS1) :: [])
  (quote S1*) []
  (quote indS1*)
  (quote S1) S1points S1paths
```

The primitive `unquoteDecl` declares β_{indS_1} as a postulate. β_{indS_1} gives the action of dependent function $indS_1$ on the path constructor `loop`.

5 Application

The field of homotopy type theory is less well-developed on the programming side. There are only few programming applications of homotopy type theory, and the role of computationally relevant equality proofs on programming is an area of active research. Applications such as homotopical patch theory [4] discuss the implementation of Darcis [16] version control system using patch theory [12] [9] in the context of homotopy type theory. Containers in homotopy type theory [3] [2] implement data structures such as multisets and cycles. The automation tool discussed in this paper abstracts away the difficulties involved in the implementation of a higher inductive type and its elimination rules. It introduces interfaces which simplify the intricacies of a higher inductive type definition and usage by automating the generation of the code segments defining the higher inductive type and its elimination rules. The automation tool is significant in reducing the development effort for existing applications, and it can also attract new programming applications in homotopy type theory.

In the next section, we discuss the implementation of patch theory application in Agda and exhibit a tremendous reduction in code size using the automation tool. In section 5.2, we present a cryptography application and discuss how to abstract the implementation difficulties of a higher inductive type making it more accessible to the cryptographers.

5.1 Patch Theory Revisited

A patch is a syntactic representation of a function that modifies a repository context when applied. For example, a patch $(s1 \leftrightarrow s2 @ l)$, which replaces string $s1$ with $s2$ at line l , when applied to a repository context with string $s1$ at line l results in a repository context with string $s2$ at line l . In homotopical patch theory [4], the patches are modeled as paths in a higher inductive type. The higher inductive type representation of patches automatically satisfy groupoid laws such as the composition of patches is associative, and inverse composes to identity. Domain-specific laws related to the patches such as two swaps at independent lines commute are designed as higher dimensional paths. The computation content of the patches is extracted by mapping them to bijections in the universe with the help of univalence. Due to the functoriality of mappings in type theory, the functions preserve the path structures in their mapping to the universe.

We developed the patch theory application in Agda using Dan Licata's method [11]. We implemented basic patches like the insertion of a string as line $l1$ in a file and deletion of a line $l2$ from a file. The functions implementing insertion and deletion in the universe are not bijective. So, to map the paths representing the patches insert and delete into the universe,

we used the patch history approach [4]. According to this approach, we developed a separate higher inductive type *History* which serves as the types of patches. In addition to basic patches, we also implemented patches of encryption using cryptosystems like rsa [15] and paillier [13].

We used the automation tool described in this paper to generate code for the higher inductive type definition representing *History* and the repository context *cryptR* for the patches. We also automated the code generation for the elimination and the computation rules for the higher inductive types *History* and *cryptR*. In addition to abstracting the implementation difficulties of higher inductive types, the automation tool helped us to achieve an extensive reduction in the code size of the original application. We were able to automate the generation of approximately 1500 lines of code with just about 70 lines of automation code. The automation massively reduced the code size of the application which is about 2500 lines resulting in 60% reduction in the original code size.

5.2 Cryptography

The work of [18] applies the tools of homotopy type theory for cryptographic protocol implementation. It introduces a new approach for the formal specification of cryptographic schemes using types. The work discusses modeling *cryptDB* [14] using a framework similar to patch theory. *CryptDB* employs layered encryption techniques and demonstrates computation on top of encrypted data. We can implement *cryptDB* by modeling the database queries as paths in a higher inductive type and mapping the paths to the universe using singleton types [4]. The automation tool can be applied to generate code for the higher inductive type representing *cryptDB* and its corresponding elimination and computation rules. By using the automation tool, we can abstract the convolutions of homotopy type theory thus making it more accessible to the broad community of cryptography.

A formal specification of a cryptographic construction promises correctness of properties related to security and implementation. The downside of formal specification is that it introduces a framework which requires expert knowledge on theorem proving and a strong mathematical background. By automating the code constructions for the mathematical part such as the higher inductive type implementation, we simplify theorem proving and formal specification to a considerable extent and make it more accessible to regular programmers without a strong mathematical background.

6 Conclusion and Future Work

We presented an automation tool developed using the new reflection library of Agda extended with support for elaborator reflection. Our automation tool handles code generation for inductive types with constructors taking zero arguments, one or more arguments, and type being defined itself as an

argument. We simplified the syntax for defining higher inductive types through the mechanized construction of the boiler-plate code segments. By automating the generation of the elimination and the computation rules associated with a higher inductive type, we demonstrated an extensive reduction in code size and abstraction of difficulties involved in implementing and using the higher inductive type. Next, we intend to extend the support to include more categories of the inductive type such as the inductive-inductive type and the inductive-recursive type.

References

- [1] 2017. *Agda's Documentation*. <http://agda.readthedocs.io/en/latest/language/reflection.html>.
- [2] Michael Abbott, Thorsten Altenkirch, and Neil Ghani. 2005. Containers: constructing strictly positive types. *Theoretic Computer Science*.
- [3] Thorsten Altenkirch. 2014. Containers in homotopy type theory. (January 2014). Talk at Mathematical Structures of Computation, Lyon.
- [4] Carlo Angiuli, Edward Morehouse, Daniel R. Licata, and Robert Harper. 2014. Homotopical Patch Theory. In *Proceedings of the 19th ACM SIGPLAN International Conference on Functional Programming (ICFP '14)*. ACM.
- [5] Ana Bove and Venanzio Capretta. 2005. Modelling general recursion in type theory. *Mathematical Structures in Computer Science*, 15(4):671–708.
- [6] David Christiansen. 2005. Dependent type providers. In *Proceedings of the 9th ACM SIGPLAN Workshop on Generic Programming, WGP '13*, New York, USA.
- [7] David Christiansen. 2016. *Practical Reflection and Metaprogramming for Dependent Types*. Ph.D. Dissertation. IT University of Copenhagen.
- [8] David Christiansen and Edwin Brady. 2016. Elaborator Reflection: Extending Idris in Idris. In *Proceedings of the 21st ACM SIGPLAN International Conference on Functional Programming (ICFP '16)*. ACM, Nara, Japan.
- [9] Jason Dagit. 2009. Type-correct changes—a safe approach to version control implementation. (2009). MS Thesis.
- [10] Gabriel Ebner, Sebastian Ullrich, Jared Roesch, Jeremy Avigad, and Leonardo de Moura. 2017. A Metaprogramming Framework for Formal Verification. *Proceedings of the ACM on Programming Languages* 1, ICFP, Article 34 (August 2017), 29 pages.
- [11] Daniel R. Licata. 2011. Running Circles Around (In) Your Proof Assistant; or, Quotients that Compute. (April 2011). <https://homotopytypetheory.org/2011/04/23/running-circles-around-in-your-proof-assistant>.
- [12] Samuel Mimram and Cinzia Di Giusto. 2013. A categorical theory of patches. *Electronic Notes in Theoretic Computer Science*, 298:283–307.
- [13] Pascal Paillier. 1999. Public-key cryptosystems based on composite degree residuosity classes. In: *Proceedings of the 18th Annual International Conference on the Theory and Applications of Cryptographic Techniques (EUROCRYPT)*, Prague, Czech Republic.
- [14] Raluca Ada Popa, Catherine M.S. Redfield, and Hari Balakrishnan Nickolai Zeldovich. 2011. *CryptDB: Protecting confidentiality with encrypted query processing*. In: *Proceedings of the 23rd ACM Symposium on Operating Systems Principles (SOSP)*, Cascais, Portugal.
- [15] Adi Shamir Ron Rivest and Leonard Adleman. 1978. A Method for Obtaining Digital Signatures and Public-Key Cryptosystems. *Communications of the ACM*, 21 (2), pp. 120-126.
- [16] David Roundy. 2005. Darcs: Distributed version management in haskell. In *ACM SIGPLAN Workshop on Haskell*. ACM.
- [17] The Univalent Foundations Program. 2013. *Homotopy Type Theory: Univalent Foundations of Mathematics*. <https://homotopytypetheory>.

org/book, Institute for Advanced Study.

- [18] Paventhan Vivekanandan. 2018. A Homotopical Approach to Cryptography. (July 2018). To be presented at Workshop on Foundations of Computer Security (FCS 2018), University of Oxford, UK.